

IN-64
160258
P.17

A Brief Description of a New Numerical Framework for Solving Conservation Laws—The Method of Space-Time Conservation Element and Solution Element

Sin-Chung Chang
National Aeronautics and Space Administration
Lewis Research Center
Cleveland, Ohio

and

Wai-Ming To
Sverdrup Technology, Inc.
Lewis Research Center Group
Brook Park, Ohio

Prepared for the
13th International Conference on Numerical Methods
in Fluid Dynamics
Rome, Italy, July 6–10, 1992



(NASA-TM-105757) A BRIEF
DESCRIPTION OF A NEW NUMERICAL
FRAMEWORK FOR SOLVING CONSERVATION
LAWS: THE METHOD OF SPACE-TIME
CONSERVATION ELEMENT AND SOLUTION
ELEMENT (NASA) 17 p

N93-26560

Unclass

G3/64 0160258

A BRIEF DESCRIPTION OF A NEW NUMERICAL FRAMEWORK FOR SOLVING CONSERVATION LAWS — THE METHOD OF SPACE-TIME CONSERVATION ELEMENT AND SOLUTION ELEMENT §

S.C. Chang † and W.M. To ‡

† NASA Lewis Research Center, Cleveland, Ohio 44135, U.S.A.

‡ Sverdrup Technology, Inc., Cleveland, Ohio 44142, U.S.A.

1. INTRODUCTION

In this paper we shall describe a new numerical method for solving conservation laws. It is much simpler than a typical high resolution method [1]. No flux limiter or any characteristics-based technique is involved. No artificial viscosity or smoothing is introduced, and no moving mesh is used. Yet this method is capable of generating highly accurate shock tube solutions. The slight numerical overshoot and/or oscillations generated can be removed if a simple averaging formula initially used is replaced by a weighted averaging formula. This modification has no discernable effect on other parts of the solution. Because of its simplicity, multi-dimension generalization is straightforward and it allows for the simultaneous treatment of variables in different spatial directions.

2. CONSERVATION LAWS

We consider a dimensionless form of the 1-D unsteady Euler equations for an ideal gas. Let ρ , u , p , and γ , respectively, be the mass density, velocity, static pressure, and constant specific heat ratio. Let

$$q_1 = \rho \quad , \quad q_2 = \rho u \quad , \quad q_3 = p / (\gamma - 1) + (1/2) \rho u^2 \quad (2.1)$$

$$\left. \begin{aligned} f_1 &= q_2 \\ f_2 &= (\gamma - 1) q_3 + (1/2) (3 - \gamma) (q_2)^2 / q_1 \\ f_3 &= \gamma q_2 q_3 / q_1 - (1/2) (\gamma - 1) (q_2)^3 / (q_1)^2 \end{aligned} \right\} \quad (2.2)$$

Then the Euler equations can be expressed as

$$\partial q_m / \partial t + \partial f_m / \partial x = 0 \quad , \quad m = 1, 2, 3 \quad (2.3)$$

§ This work is dedicated to the memory of a teacher, Mr. Nylon Cheng

Let $x_1 = x$ and $x_2 = t$ be considered as the coordinates of a two-dimensional Euclidean space E_2 . The integral form of Eq. (2.3) in the space-time E_2 can be expressed as (see Fig. 1)

$$\oint_{S(V)} \vec{h}_m \cdot \vec{ds} = 0 \quad , \quad m = 1, 2, 3 \quad (2.4)$$

where (i) $S(V)$ is the boundary of an arbitrary space-time volume V in E_2 , (ii) $\vec{h}_m = (f_m, q_m)$ are space-time current density vectors in E_2 , and (iii) $\vec{ds} = d\sigma \vec{n}$ with $d\sigma$ and \vec{n} , respectively, being the area and the outward unit normal of a surface element on $S(V)$. Note that (i) $\vec{h}_m \cdot \vec{ds}$ is the space-time flux of \vec{h}_m leaving the volume V through the surface element \vec{ds} , and (ii) all mathematical operations can be carried out as though E_2 is an ordinary two-dimensional Euclidean space.

3. NUMERICAL METHOD

Let E_2 be divided into nonoverlapping rhombic regions (see Fig. 2) referred to as solution elements (SEs). Each SE is centered at a mesh point (j, n) where $n = 0, 1/2, 1, 3/2, \dots$, and $j = (n \pm 1/2), (n \pm 3/2), \dots$, i.e., j is a half-integer (whole integer) if n is a whole integer (half-integer). In other words, j and n are both whole integers or both half-integers if, as occurring in Fig. 2, a SE is centered at the mesh point $(j, n+1/2)$. Thus, the locations of SEs and their centers are staggered over every half time-step. A SE centered at (j, n) , and its interior are denoted by $SE(j, n)$ and $SE'(j, n)$, respectively.

For any $(x, t) \in SE'(j, n)$, $q_m(x, t)$, $f_m(x, t)$, and $\vec{h}_m(x, t)$, respectively, are approximated by $\underline{q}_m(x, t; j, n)$, $\underline{f}_m(x, t; j, n)$, and $\vec{\underline{h}}_m(x, t; j, n)$ which we shall define immediately. Let

$$\underline{q}_m(x, t; j, n) = (\sigma_m)_j^n + (\alpha_m)_j^n (x - x_j) + (\beta_m)_j^n (t - t^n) \quad , \quad m = 1, 2, 3 \quad (3.1)$$

where $(\sigma_m)_j^n$, $(\alpha_m)_j^n$, and $(\beta_m)_j^n$ are constants in $SE'(j, n)$, and (x_j, t^n) are the coordinates of the mesh point (j, n) . Note that

$$\underline{q}_m(x_j, t^n; j, n) = (\sigma_m)_j^n \quad , \quad \partial \underline{q}_m(x, t; j, n) / \partial x = (\alpha_m)_j^n \quad , \quad \partial \underline{q}_m(x, t; j, n) / \partial t = (\beta_m)_j^n \quad (3.2)$$

Moreover, if we identify $(\sigma_m)_j^n$, $(\alpha_m)_j^n$, and $(\beta_m)_j^n$, respectively, with the values of q_m , $\partial q_m / \partial x$, and $\partial q_m / \partial t$ at (x_j, t^n) , the expression on the right side of Eq. (3.1) becomes the first-order Taylor's expansion of $q_m(x, t)$ at (x_j, t^n) . As a result of these considerations, $(\sigma_m)_j^n$, $(\alpha_m)_j^n$, and $(\beta_m)_j^n$ will be considered as the numerical analogues of the values of q_m , $\partial q_m / \partial x$, and $\partial q_m / \partial t$ at (x_j, t^n) , respectively.

Let $\underline{f}_m(x, t; j, n)$, $m = 1, 2, 3$, be defined in terms of $\underline{q}_m(x, t; j, n)$, $m = 1, 2, 3$, according to Eq. (2.2) with the understanding that f_m and q_m in Eq. (2.2) be replaced by $\underline{f}_m(x, t; j, n)$ and $\underline{q}_m(x, t; j, n)$, respectively. Using Eq. (3.1), $\underline{f}_m(x, t; j, n)$ is expressed as a function of $(x - x_j)$ and $(t - t^n)$, and then expanded as a power series of them. The new method is simplified by truncating

the power series after first-order terms. This is consistent with the first-order approximation given in Eq. (3.1). In the new method, only f_m at $x = x_j$ are needed. Let $\sigma_m = (\sigma_m)_j^n$ and $\beta_m = (\beta_m)_j^n$. Then explicitly, we have:

$$f_1(x_j, t; j, n) = \sigma_2 + \beta_2(t-t^n) \quad (3.3)$$

$$f_2(x_j, t; j, n) = (\gamma-1)\sigma_3 + (1/2)(3-\gamma)(\sigma_2)^2/\sigma_1 \\ + [(\gamma-1)\beta_3 + (3-\gamma)(\sigma_2\beta_2/\sigma_1) - (1/2)(3-\gamma)(\sigma_2/\sigma_1)^2\beta_1](t-t^n) \quad (3.4)$$

$$f_3(x_j, t; j, n) = \gamma\sigma_2\sigma_3/\sigma_1 - (1/2)(\gamma-1)(\sigma_2)^3/(\sigma_1)^2 + \{\gamma[(\sigma_2\beta_3 + \sigma_3\beta_2)/\sigma_1 \\ - \sigma_2\sigma_3\beta_1/(\sigma_1)^2] + (1/2)(\gamma-1)[2(\sigma_2/\sigma_1)^3\beta_1 - 3(\sigma_2/\sigma_1)^2\beta_2]\}(t-t^n) \quad (3.5)$$

Since $\vec{h}_m = (f_m, q_m)$, we define $\vec{h}_m(x, t; j, n) = (f_m(x, t; j, n), q_m(x, t; j, n))$.

Let E_2 be divided into nonoverlapping rectangular regions (see Fig. 2) referred to as conservation elements (CEs). They are also staggered over every half time-step. Let the CE with its upper edge centered at (j, n) be denoted by $CE(j, n)$. Then the current approximation of Eq. (2.4) is

$$\oint_{S(CE(j, n))} \vec{h}_m \cdot \vec{ds} = 0 \quad (\text{all possible } m \text{ and } (j, n)) \quad (3.6)$$

Because the entire boundary (except for three isolated points) of a CE is located within the interiors of three neighboring SEs, \vec{h}_m is continuous across any interface separating two neighboring CEs. Thus Eq. (3.6) will remain valid if $CE(j, n)$ is replaced by the union of any combination of CEs.

Because each $S(CE(j, n))$ is a simple closed curve in E_2 (see Fig. 1), the surface integration form Eq. (3.6) can be converted into a line integration form [2, p.14], i.e.,

$$\oint_{S(CE(j, n))} \vec{g}_m \cdot \vec{dr} = 0 \quad (\text{all possible } m \text{ and } (j, n)) \quad (3.7)$$

where $\vec{g}_m = (-q_m, f_m)$ and $\vec{dr} = (dx, dt)$.

For each $SE(j, n)$, let

$$s_1(j, n) = (\Delta x/8)\alpha_1 + (1/2)(\Delta t/\Delta x)[\sigma_2 + (\Delta t/4)\beta_2] \quad (3.8)$$

$$s_2(j, n) = (\Delta x/8)\alpha_2 + (1/2)(\Delta t/\Delta x)[(\gamma-1)\sigma_3 + (1/2)(3-\gamma)(\sigma_2)^2/\sigma_1] \\ + (1/8)[(\Delta t)^2/\Delta x][(\gamma-1)\beta_3 + (3-\gamma)[\sigma_2\beta_2/\sigma_1 - (1/2)(\sigma_2/\sigma_1)^2\beta_1]] \quad (3.9)$$

$$s_3(j, n) = (\Delta x/8)\alpha_3 + (1/2)(\Delta t/\Delta x)[\gamma\sigma_2\sigma_3/\sigma_1 - (1/2)(\gamma-1)(\sigma_2)^3/(\sigma_1)^2] + (1/8)[(\Delta t)^2/\Delta x] \\ \times \{(\gamma/\sigma_1)(\sigma_2\beta_3 + \sigma_3\beta_2 - \sigma_2\sigma_3\beta_1/\sigma_1) + (\gamma-1)[(\sigma_2/\sigma_1)^3\beta_1 - (3/2)(\sigma_2/\sigma_1)^2\beta_2]\} \quad (3.10)$$

where $\sigma_m = (\sigma_m)_j^n$, $\alpha_m = (\alpha_m)_j^n$, and $\beta_m = (\beta_m)_j^n$. Then Eq. (3.7) implies that, for each $SE(j, n+1/2)$,

$$(\sigma_m)_j^{n+1/2} = (1/2)[(\sigma_m)_{j-1/2}^n + (\sigma_m)_{j+1/2}^n] + s_m(j-1/2, n) - s_m(j+1/2, n) \quad , \quad m = 1, 2, 3 \quad (3.11)$$

i.e., $(\sigma_m)_j^{n+1/2}$ is determined in terms of the numerical variables associated with $SE(j-1/2, n)$ and $SE(j+1/2, n)$. Similar formulae for $(\alpha_m)_j^{n+1/2}$ and $(\beta_m)_j^{n+1/2}$ will be given next.

Let A_+ , A , and A_- (see Fig. 2) denote $(x_{j+1/2}, t^{n+1/2})$, $(x_j, t^{n+1/2})$, and $(x_{j-1/2}, t^{n+1/2})$, respectively. Let

$$q_{m\pm} = q_m(x_{j\pm 1/2}, t^{n+1/2}; j\pm 1/2, n) \quad (3.12)$$

Because A_{\pm} do not belong to $SE'(j\pm 1/2, n)$, the expression on the right side of Eq. (3.12) is to be evaluated at the two points immediately below them. A central-difference formula for evaluating $(\alpha_m)_j^{n+1/2}$, the numerical analogue of $\partial q_m / \partial x$ at point A , is

$$(\alpha_m)_j^{n+1/2} = (q_{m+} - q_{m-}) / \Delta x \quad (3.13)$$

This formula is valid as long as no discontinuity of q_m (or its derivatives) occurs between A_- and A_+ . In the following discussion, we develop an alternate which is valid even in the presence of discontinuity.

Let

$$(\alpha_{m\pm})_j^{n+1/2} = \pm [q_{m\pm} - (\sigma_m)_j^{n+1/2}] / (\Delta x / 2) \quad (3.14)$$

where $(\sigma_m)_j^{n+1/2}$ has just been obtained using Eq. (3.11). Because q_{m+} , $(\sigma_m)_j^{n+1/2}$, and q_{m-} are the numerical analogues of q_m at A_+ , A , and A_- , respectively, $(\alpha_{m+})_j^{n+1/2}$ and $(\alpha_{m-})_j^{n+1/2}$ are two numerical analogues of $\partial q_m(x_j, t^{n+1/2}) / \partial x$ with one being evaluated from the right and another from the left. Note that $(\alpha_m)_j^{n+1/2}$ defined by Eq. (3.13) is equal to the average of $(\alpha_{m+})_j^{n+1/2}$ and $(\alpha_{m-})_j^{n+1/2}$.

In case that a discontinuity occurs between A and A_+ but not between A and A_- , one would expect that $|(\alpha_{m+})_j^{n+1/2}| \gg |(\alpha_{m-})_j^{n+1/2}|$. Moreover, because A and A_- are on the same side of the discontinuity while A and A_+ are on the opposite sides, $(\alpha_m)_j^{n+1/2}$ should be closer to $(\alpha_{m-})_j^{n+1/2}$ than $(\alpha_{m+})_j^{n+1/2}$. This observation suggests that $(\alpha_m)_j^{n+1/2}$ should be a weighted average of $(\alpha_{m+})_j^{n+1/2}$ and $(\alpha_{m-})_j^{n+1/2}$ biased toward the one with the smaller magnitude.

As a result of the above and other considerations [3], Eq. (3.13) will be generalized by

$$(\alpha_m)_j^{n+1/2} = F((\alpha_{m-})_j^{n+1/2}, (\alpha_{m+})_j^{n+1/2}; c) \quad (3.15)$$

Here $c \geq 0$ is an adjustable constant and the function F is defined by (i) $F(0, 0; c) = 0$ and (ii)

$$F(\alpha_-, \alpha_+; c) = (|\alpha_+|^c \alpha_- + |\alpha_-|^c \alpha_+) / (|\alpha_+|^c + |\alpha_-|^c) \quad , \quad (|\alpha_+| + |\alpha_-| > 0) \quad (3.16)$$

where α_- and α_+ are any two real variables. Note that $F(\alpha_-, \alpha_+; c) = (\alpha_- + \alpha_+) / 2$ if $c = 0$ or $|\alpha_-| = |\alpha_+|$, i.e., Eq. (3.15) is reduced to Eq. (3.13) if $c = 0$ or $|\alpha_-| = |\alpha_+|$. Also the expression on the right side of Eq. (3.16) represents a weighted average of α_- and α_+ with the weight factors $|\alpha_+|^c / (|\alpha_+|^c + |\alpha_-|^c)$ and $|\alpha_-|^c / (|\alpha_+|^c + |\alpha_-|^c)$. For $c > 0$, this average is biased toward the one among α_- and α_+ with the smaller magnitude. For the same values of $|\alpha_+|$ and $|\alpha_-|$, the bias increases as c increases.

Substituting Eq. (2.2) into Eq. (2.3), one obtains three equations in which $\partial q_m / \partial t$, $m = 1, 2, 3$, are expressed in terms of q_m and $\partial q_m / \partial x$, $m = 1, 2, 3$. Let q_m and their derivatives be replaced by the corresponding numerical analogues at the mesh point $(j, n+1/2)$, one obtains that

$$(\beta_1)_j^{n+1/2} = -\alpha_2 \quad (3.17)$$

$$(\beta_2)_j^{n+1/2} = (1/2)(3-\gamma)(\sigma_2/\sigma_1)^2 \alpha_1 - (3-\gamma)(\sigma_2/\sigma_1) \alpha_2 - (\gamma-1) \alpha_3 \quad (3.18)$$

$$\begin{aligned} (\beta_3)_j^{n+1/2} = & [\gamma \sigma_2 \sigma_3 / (\sigma_1)^2 - (\gamma-1)(\sigma_2/\sigma_1)^3] \alpha_1 \\ & + [(3/2)(\gamma-1)(\sigma_2/\sigma_1)^2 - \gamma \sigma_3 / \sigma_1] \alpha_2 - \gamma(\sigma_2/\sigma_1) \alpha_3 \end{aligned} \quad (3.19)$$

where $\sigma_m = (\sigma_m)_j^{n+1/2}$ and $\alpha_m = (\alpha_m)_j^{n+1/2}$.

With the aid of Eqs. (3.11), (3.15), and (3.17) - (3.19), $(\sigma_m)_j^n$, $(\alpha_m)_j^n$, and $(\beta_m)_j^n$ can be determined in terms of the initial values $(\sigma_m)_{\pm 1/2}^0$, $(\sigma_m)_{\pm 3/2}^0$, \dots , and $(\alpha_m)_{\pm 1/2}^0$, $(\alpha_m)_{\pm 3/2}^0$, \dots .

4. NUMERICAL RESULTS

We consider a shock tube problem used by Sod [4]. Let $\gamma = 1.4$. At $t = 0$, let (i) $(\rho, u, p) = (1, 0, 1)$, i.e., $(q_1, q_2, q_3) = (1, 0, 2.5)$, if $x < 0$, and (ii) $(\rho, u, p) = (0.125, 0, 0.1)$, i.e., $(q_1, q_2, q_3) = (0.125, 0, 0.25)$, if $x > 0$. Thus

$$(i) \quad ((\sigma_1)_j^0, (\sigma_2)_j^0, (\sigma_3)_j^0) = \begin{cases} (1, 0, 2.5) & \text{if } j = -1/2, -3/2, \dots \\ (0.125, 0, 0.25) & \text{if } j = 1/2, 3/2, \dots \end{cases} \quad (4.1)$$

and (ii) $(\alpha_m)_j^0 = 0$, $j = \pm 1/2, \pm 3/2, \dots$. Eqs. (3.17) - (3.19) imply that $(\beta_m)_j^0 = 0$, $j = \pm 1/2, \pm 3/2, \dots$.

The above initial conditions, and Eqs. (3.11), (3.15), and (3.17) - (3.19), imply that $(\sigma_m)_j^n$, $(\alpha_m)_j^n$, and $(\beta_m)_j^n$ are constant in two separate regions which, respectively, are defined by $j \leq -(n+1/2)$ and $j \geq (n+1/2)$. Thus one needs to evaluate the above variables only if $(n+1/2) > |j|$.

The current scheme is stable if $CFL = \max(|u| + |a|)\Delta t / \Delta x \leq 1$ [3]. Here a = local sound speed. In the current computations, $\Delta x = 0.01$, $\Delta t = 0.004$, and $CFL = 0.88$. Numerical results (dots) at $t = 0.4$ are compared with the exact solutions (solid lines) in Fig. 3. Since each

marching step advances the solution from t to $t+\Delta t/2$, these results are obtained after 200 steps. Note that (i) shock discontinuity is resolved almost within one mesh interval, and (ii) the slight numerical overshoot and oscillations generated when $c = 0$ essentially disappear when $c = 1$ is used.

5. CONCLUSIONS AND DISCUSSIONS

The current scheme has a stencil containing only two points. This minimization of stencil has the effect of reducing numerical diffusion [5]. It is achieved by including $(\alpha_m)_j^n$ and $(\beta_m)_j^n$ as numerical variables. The fluxes at an interface separating two CEs are evaluated with no interpolation or extrapolation. Accuracy of flux evaluation is enhanced by requiring that the solution given in Eq. (3.1) satisfies the Euler equations at the center of every SE. This makes the use of characteristics-based techniques less necessary. The above key features all contribute to the simplicity, generality, and accuracy of the current scheme. They all owe their existence to the use of staggered SEs and CEs.

In the current method, flux evaluation within each $SE(j,n)$ is required only at a subset of $SE(j,n)$, i.e., a horizontal line segment centered at (j,n) and a vertical line segment starting upward from (j,n) (see Fig. 2). As a result, we may redefine $SE(j,n)$ to be this subset. This new definition is used in the following sketch of an extension of the the current scheme to a three-dimensional Euclidean space E_3 ($x_1 = x$, $x_2 = y$, and $x_3 = t$).

A SE contains three mutually perpendicular rectangles (see Fig. 4a). The point of intersection is referred to as the center of this SE. The SEs are staggered in both x - and y - directions over every half time-step. The CEs are rectangular boxes (see Fig. 4b) also staggered in both x - and y - directions over every half time-step. From Fig. 4b, it is seen that the boundary of a CE can be divided into five parts which, respectively, belong to five neighboring SEs. As a result, the solution procedure described in Section 3 can be easily extended to E_3 .

REFERENCES

1. R.J. LeVeque, *Numerical Methods for Conservation Laws*, Birkhäuser Verlag, 1990.
2. S.C. Chang and W.M. To, *A New Numerical Framework for Solving Conservation Laws — The Method of Space-Time Conservation Element and Solution Element*, NASA TM 104495.
3. S.C. Chang and W.M. To, *A New Development in the Method of Space-Time Conservation Element and Solution Element — Application to Shock Tube Problem*, to appear.
4. G.A. Sod, *J. Computational Phys.* 27 (1978), 1.
5. S.C. Chang, *On An Origin of Numerical Diffusion: Violation of Invariance under Space-Time Inversion*, to appear in the proceeding of 23rd Conference on Modeling and Simulation, April 30 – May 1, 1992, Pittsburgh, PA, USA.




Figure 1.—A surface element $d\vec{s}$ and a line segment $d\vec{r}$ on the boundary $S(V)$ of a volume V in a space-time E_2 .




Figure 2.—The SEs (the rhombuses formed by solid lines) and the CEs (the rectangles formed by dashed lines). The dots represent the centers of SEs.






Figure 3. Shock tube solution at $t = 0.4$.

(d) Pressure ($c = 1$)



(e) Density ($c = 1$)



(f) Velocity ($c = 1$)






Figure 3 (cont.). Shock tube solution at $t = 0.4$.

(g) Pressure ($c = 2$)



(h) Density ($c = 2$)



(i) Velocity ($c = 2$)




Figure 3 (cont.). Shock tube solution at $t = 0.4$.






Figure 3 (cont.). Shock tube solution at $t = 0.4$.




Figure 4.—A SE and a CE in E_3 . The dots represent the centers of SEs.

```

implicit real*8(a-h,o-z)
c
dimension q1(1000), q2(1000), q3(1000), q1x(1000), q2x(1000),
*         q3x(1000), q1t(1000), q2t(1000), q3t(1000), q1n(1000),
*         q2n(1000), q3n(1000), s1(1000), s2(1000), s3(1000),
*         xx(1000)
c
it = 200
iw = 200
iwi = 10
dt = 0.4d-2
dx = 0.1d-1
ga = 1.4d0
rhol = 1.d0
ul = 0.d0
pl = 1.d0
rhor = 0.125d0
ur = 0.d0
pr = 0.1d0
ic = 1
c
open (unit=8,file='for008')
write (8,10) it,iw,iwi,ic
write (8,20) dt,dx,ga
write (8,30) rhol,ul,pl
write (8,40) rhor,ur,pr
c
dt1 = 0.5d0*dt
dt2 = dt**2/8.d0
hdx = 0.5d0*dx
dx1 = 0.25d0*dx
dx2 = 0.125d0*dx
ga1 = ga - 1.d0
ga2 = 3.d0 - ga
ga3 = 0.5d0*ga2
ga4 = 0.5d0*ga1
q1(1) = rhol
q2(1) = rhol*ul
q3(1) = pl/ga1 + 0.5d0*rhol*ul**2
q1(2) = rhor
q2r = rhor*ur
q3r = pr/ga1 + 0.5d0*rhor*ur**2
q2(2) = q2r
q3(2) = q3r
q1x(1) = 0.d0
q1x(2) = 0.d0
q2x(1) = 0.d0
q2x(2) = 0.d0
q3x(1) = 0.d0
q3x(2) = 0.d0
q1t(1) = 0.d0
q1t(2) = 0.d0
q2t(1) = 0.d0
q2t(2) = 0.d0
q3t(1) = 0.d0
q3t(2) = 0.d0
c
m = 2
do 600 i = 1,it
c

```

```

do 100 j = 1,m
v1 = q1(j)
v2 = q2(j)
v3 = q3(j)
v1t = q1t(j)
v2t = q2t(j)
v3t = q3t(j)
c
s1(j) = dx2*q1x(j) + (dt1*v2 + dt2*v2t)/dx
c
s2(j) = dx2*q2x(j) + (dt1*(ga1*v3 + (ga3/v1)*v2**2) + dt2*(ga1*v3t
*      + (ga3/v1)*(2.d0*v2*v2t - v2**2*v1t/v1)))/dx
c
s3(j) = dx2*q3x(j) + (dt1*(ga*v2*v3/v1 - ga4*v2**3/v1**2) + dt2*
*      ((ga/v1)*(v2*v3t + v3*v2t - v2*v3*v1t/v1) - (ga4*v2/v1**2)
*      *(3.d0*v2*v2t - 2.d0*v2**2*v1t/v1)))/dx
c
100 continue
c
mm = m - 1
do 200 j = 1,mm
q1n(j+1) = 0.5d0*(q1(j) + q1(j+1)) + s1(j) - s1(j+1)
q2n(j+1) = 0.5d0*(q2(j) + q2(j+1)) + s2(j) - s2(j+1)
q3n(j+1) = 0.5d0*(q3(j) + q3(j+1)) + s3(j) - s3(j+1)
v1x1 = (q1n(j+1) - q1(j) - dt1*q1t(j))/hdx
v1xr = (q1(j+1) + dt1*q1t(j+1) - q1n(j+1))/hdx
v2x1 = (q2n(j+1) - q2(j) - dt1*q2t(j))/hdx
v2xr = (q2(j+1) + dt1*q2t(j+1) - q2n(j+1))/hdx
v3x1 = (q3n(j+1) - q3(j) - dt1*q3t(j))/hdx
v3xr = (q3(j+1) + dt1*q3t(j+1) - q3n(j+1))/hdx
q1x(j+1) = (v1x1*(dabs(v1xr)**ic + v1xr*(dabs(v1x1))**ic)/
*      ((dabs(v1x1))**ic + (dabs(v1xr))**ic + 1.d-60)
q2x(j+1) = (v2x1*(dabs(v2xr))**ic + v2xr*(dabs(v2x1))**ic)/
*      ((dabs(v2x1))**ic + (dabs(v2xr))**ic + 1.d-60)
q3x(j+1) = (v3x1*(dabs(v3xr))**ic + v3xr*(dabs(v3x1))**ic)/
*      ((dabs(v3x1))**ic + (dabs(v3xr))**ic + 1.d-60)
200 continue
c
do 300 j = 2,m
q1(j) = q1n(j)
q2(j) = q2n(j)
q3(j) = q3n(j)
v1 = q1(j)
v2 = q2(j)
v3 = q3(j)
v1x = q1x(j)
v2x = q2x(j)
v3x = q3x(j)
q1t(j) = -v2x
q2t(j) = ga3*(v2/v1)**2*v1x - ga2*(v2/v1)*v2x - ga1*v3x
q3t(j) = (ga*v2*v3/v1**2 - ga1*(v2/v1)**3)*v1x + (1.5d0*ga1*
*      (v2/v1)**2 - ga*v3/v1)*v2x - ga*(v2/v1)*v3x
300 continue
c
m = m + 1
q1(m) = rhor
q2(m) = q2r
q3(m) = q3r
q1x(m) = 0.d0
q2x(m) = 0.d0

```

```

q3x(m) = 0.d0
q1t(m) = 0.d0
q2t(m) = 0.d0
q3t(m) = 0.d0
c
  if (i.ne.iw) goto 600
  iw = iw + iwi
  t = dt1*dfloat(i)
  write (8,50) i,t
  mm = m - 1
  t2 = dx*dfloat(mm)
  xx(1) = -0.5d0*t2
c
  do 400 j = 1,mm
  xx(j+1) = xx(j) + dx
400  continue
c
  do 500 j = 1,m
  x = q2(j)/q1(j)
  y = q3(j)/q1(j) - 0.5d0*x**2
  z = gal*y*q1(j)
  write (8,60) xx(j),y,q1(j),x,z
500  continue
600  continue
c
  close (unit=8)
10  format(' it = ',i4,' iw = ',i4,' iwi = ',i4,' ic = ',i4)
20  format(' dt = ',g14.7,' dx = ',g14.7,' gamma = ',g14.7)
30  format(' rhol = ',g14.7,' ul = ',g14.7,' pl = ',g14.7)
40  format(' rhor = ',g14.7,' ur = ',g14.7,' pr = ',g14.7)
50  format(' i = ',i4,' t = ',g14.7,'*****')
60  format(' x = ',f8.4,' e = ',g11.4,' rho = ',g11.4,' u = ',g11.4,
*      ' p = ',g11.4)
  stop
  end

```

REPORT DOCUMENTATION PAGE

Form Approved
OMB No. 0704-0188

Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project (0704-0188), Washington, DC 20503.

1. AGENCY USE ONLY (Leave blank)	2. REPORT DATE July 1992	3. REPORT TYPE AND DATES COVERED Technical Memorandum	
4. TITLE AND SUBTITLE A Brief Description of a New Numerical Framework for Solving Conservation Laws—The Method of Space-Time Conservation Element and Solution Element		5. FUNDING NUMBERS WU-505-62-52	
6. AUTHOR(S) Sin-Chung Chang and Wai-Ming To		7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) National Aeronautics and Space Administration Lewis Research Center Cleveland, Ohio 44135-3191	
8. PERFORMING ORGANIZATION REPORT NUMBER E-7169		9. SPONSORING/MONITORING AGENCY NAMES(S) AND ADDRESS(ES) National Aeronautics and Space Administration Washington, D.C. 20546-0001	
10. SPONSORING/MONITORING AGENCY REPORT NUMBER NASA TM-105757		11. SUPPLEMENTARY NOTES Prepared for the 13th International Conference on Numerical Methods in Fluid Dynamics, Rome, Italy, July 6-10, 1992. Sin-Chung Chang, NASA Lewis Research Center, Cleveland, Ohio 44135, and Wai-Ming To, Sverdrup Technology, Inc., Lewis Research Center Group, 2001 Aerospace Parkway, Brook Park, Ohio 44142. Responsible person, Sin-Chung Chang, (216) 433-5874.	
12a. DISTRIBUTION/AVAILABILITY STATEMENT Unclassified - Unlimited Subject Category 64		12b. DISTRIBUTION CODE	
13. ABSTRACT (Maximum 200 words) A new numerical method for solving conservation laws is being developed. It differs substantially from the well-established methods, i.e., finite difference, finite volume, finite element, and spectral methods, in both concept and methodology. It is much simpler than a typical high resolution method. No flux limiter or any technique related to characteristics is involved. No artificial viscosity or smoothing is introduced, and no moving mesh is used. Yet this method is capable of generating highly accurate shock tube solutions. The slight numerical overshoot and/or oscillations generated can be removed if a simple averaging formula initially used is replaced by a weighted formula. This modification has little effect on other parts of the solution. Because of its simplicity, generalization of this new method for multi-dimensional problems is straightforward.			
14. SUBJECT TERMS Space-time; Conservation elements; Solution elements		15. NUMBER OF PAGES 18	
17. SECURITY CLASSIFICATION OF REPORT Unclassified		16. PRICE CODE A03	
18. SECURITY CLASSIFICATION OF THIS PAGE Unclassified	19. SECURITY CLASSIFICATION OF ABSTRACT Unclassified	20. LIMITATION OF ABSTRACT	

National Aeronautics and
Space Administration

Lewis Research Center
Cleveland, Ohio 44135

FOURTH CLASS MAIL



ADDRESS CORRECTION REQUESTED

Official Business
Penalty for Private Use \$300

NASA

